

Diffing Mutually Recursive Types

A code tour

Victor Cacciari Miraldo

University of Utrecht

November 23, 2016

1 Our Universe

The universe we are using is a variant of Regular types, but instead of having only one type variable, we handle n type variables. The codes are description of regular functors on n variables:

```
data Un (n : ℕ) : Set where
  I      : Fin n      → Un n
  K      : Fin ks#    → Un n
  u1     :             Un n
  _⊕_    : (ty tv : Un n) → Un n
  _⊗_    : (ty tv : Un n) → Un n
```

Constructor **I** refers to the n -th type variable whereas **K** refers to a constant type. Value $ks\#$ is passed as a module parameter. The denotation is defined as:

```
Parms : ℕ → Set1
Parms n = Fin n → Set

[[_]] : {n : ℕ} → Un n → Parms n → Set
[[ I x   ]] A = A x
[[ K x   ]] A = lookup x ks
[[ u1    ]] A = Unit
[[ ty ⊕ tv ]] A = [[ ty ]] A ⊔ [[ tv ]] A
[[ ty ⊗ tv ]] A = [[ ty ]] A × [[ tv ]] A
```

Note that here, **Parms** n really is isomorphic to n types that serve as the parameters to the functor $[[F]]$. When we introduce a fixpoint combinator, these parameters are used to tie the recursion knot, just like a simple fixpoint: $\mu F \equiv F(\mu F)$. In fact, a mutually recursive family can be easily encoded in this setting. All we need is n types that refer to n type-variables each!

```
Fam : ℕ → Set
Fam n = Vec (Un n) n
```

```
data Fix {n : ℕ} (F : Fam n) : Fin n → Set where
  (⟦_⟧) : ∀ {k} → [lookup k F] (Fix F) → Fix F k
```

This universe is enough to model Context-Free grammars, and hence, provides the basic barebones for diffing elements of an arbitrary programming language. In the future, it could be interesting to see what kind of diffing functionality indexed functors could provide, as these could have scoping rules and other advanced features built into them.

1.1 Agda Details

Here we clarify some Agda specific details that are agnostic to the big picture. This can be safely skipped on a first iteration.

As we mentioned above, our codes represent functors on n variables. Obviously, to program with them, we need to apply these to something. The denotation receives a function $\text{Fin } n \rightarrow \text{Set}$, denoted $\text{Parms } n$, which can be seen as a valuation for each type variable.

In the following sections, we will be dealing with values of $\llbracket ty \rrbracket_A$ for some class of valuations A , though. We need to have decidable equality for $A \ k$ and some mapping from $A \ k$ to \mathbb{N} for all k . We call such valuations a *well-behaved parameter*:

```
record WBParms {n : ℕ} (A : Parms n) : Set where
  constructor wb-parms
  field
    parm-size : ∀ {k} → A k → ℕ
    parm-cmp   : ∀ {k} (x y : A k) → Dec (x ≡ y)
```

I still have no good justification for the *parm-size* field. Later on I sketch what I believe is the real meaning of the cost function.

The following sections discuss functionality that does not depend on *parameters to codes*. We will be passing them as Agda module parameters. The first diffing technique we discuss is the trivial diff. It's module is declared as follows:

```
module RegDiff.Diff.Trivial.Base
  {ks# : ℕ} (ks : Vec Set ks#) (keqs : Vec1 Eq ks)
  {parms# : ℕ} (A : Parms parms#) (WBA : WBParms A)
  where
```

We stick to this nomenclature throughout the code. The first line handles constant types: $ks\#$ is how many constant types we have, ks is the vector of such types and $keqs$ is an indexed vector with a proof of decidable equality over such types. The second line handles type parameters: $parms\#$ is how many type-variables our codes will have, A is the valuation we are using and WBA is a proof that A is *well behaved*.

We then declare the following synonyms:

```
U : Set
U = Un parms#

sized : {p : Fin parms#} → A p → ℕ
sized = parm-size WBA

_≡-A_ : {p : Fin parms#} (x y : A p) → Dec (x ≡ y)
_≡-A_ = parm-cmp WBA

UUSet : Set1
UUSet = U → U → Set
```

2 Computing and Representing Patches

Intuitively, a *Patch* is some description of a transformation. Setting the stage, let A and B be a types, $x : A$ and $y : B$ elements of such types. A *patch* between x and y must specify a few parts:

- i) An $\text{apply}_p : A \rightarrow \text{Maybe } B$ function,
- ii) such that $\text{apply}_p x \equiv \text{just } y$.

Well, apply_p can be seen as a functional relation (R is functional iff $\text{img } R \subseteq \text{id}$) from A to B . We call this the “application” relation of the patch, and we will denote it by $p^\flat \subseteq A \times B$.

Needs discussion:

There is still a lot that could be said about this. I feel like p^b should also be invertible in the sense that:

- i) Let $(\text{inv } p)$ denote the inverse patch of p , which is a patch from B to A .
- ii) Then, $p^b \cdot (\text{inv } p)^b \subseteq \text{id}$ and $(\text{inv } p)^b \cdot p^b \subseteq \text{id}$. Assuming $(\text{inv } p)^b$ is also functional, we can use the maybe monad to represent these relations in **Set**. Writing the first equation on a diagram in **Set**, using the *apply* functions:

$$\begin{array}{ccccc}
 B & \xrightarrow{\text{apply}_{(\text{inv } p)}} & A + 1 & \xrightarrow{\text{apply}_p + \text{id}} & (B + 1) + 1 \\
 & \searrow \iota_1 & & & \downarrow \mu \\
 & & & & B + 1
 \end{array}$$

- iii) This is hard to play ball with. We want to say, in a way, that if $x (p^b) y$, then $y ((\text{inv } p)^b) x$ and vice-versa. That is, $(\text{inv } p)$ is the actual inverse of p . Using relations, one could then say that $(\text{inv } p)^b$ is the converse of $(p^b)^\circ$. That is: $(\text{inv } p)^b \equiv (p^b)^\circ$. But, if $(\text{inv } p)^b$ is functional, so is $(p^b)^\circ$. This is the same as saying that p^b is entire! If p^b is functional and entire, it is a function (and hence, total!). And that is not true.

Now, let us discuss some code and build some intuition for what is what in the above schema. We will present different parts of the code, how do they relate to this relational view and give examples here and there!

2.1 Trivial Diff

The simplest possible way to describe a transformation is to say what is the source and what is the destination of such transformation. This can be accomplished by the Diagonal functor just fine.

$$\begin{aligned}
 \Delta &: \mathbf{UUSet} \\
 \Delta \text{ ty } tv &= \llbracket \text{ty} \rrbracket A \times \llbracket \text{tv} \rrbracket A
 \end{aligned}$$

Now, take an element $(x, y) : \Delta \text{ ty } tv$. The “application” relation it defines is trivial: $\{(x, y)\}$, or, in PF style:

$$\begin{array}{ccc}
 & \xrightarrow{y \cdot \underline{x}^\circ} & \\
 \llbracket \text{ty} \rrbracket_A & \xleftarrow{\underline{x}} K \xrightarrow{\underline{y}} & \llbracket \text{tv} \rrbracket_A
 \end{array}$$

Where, for any $A, B \in \mathbf{Set}$ and $x : A$, $\underline{x} \subseteq A \times B$ represents the *everywhere* x relation, defined by

$$\underline{x} = \{(x, b) \mid b \in B\}$$

This is a horrible patch however: We can’t calculate with it because we don’t know *anything* about *how* x changed into y . Note, however, that $(x, y)^b \equiv \underline{y} \cdot \underline{x}^\circ$ is trivially functional.

2.2 Spines

We can try to make it better by identifying the longest prefix of constructors where x and y agree, before giving up and using Δ . Moreover, this becomes much easier if x and y actually have the same type. In practice, we are only interested in diffing elements of the same language. It does not make sense to diff a C source file against a Haskell source file.

Nevertheless, we define an **S** structure to capture the longest common prefix of x and y :

```

data S (P : UUSet) : U → Set where
  SX  : {ty : U} → P ty ty → S P ty
  Scp : {ty : U} → S P ty
  S⊗  : {ty tv : U}
    → S P ty → S P tv → S P (ty ⊗ tv)
  Si1 : {ty tv : U}
    → S P ty → S P (ty ⊕ tv)
  Si2 : {ty tv : U}
    → S P ty → S P (tv ⊕ ty)

```

Note that **S** makes a functor (actually, a free monad!) on P , and hence, we can map over it:

```

S-map : {ty : U} {P Q : UUSet}
  → (f : ∀ {k} → P k k → Q k k)
  → S P ty → S Q ty
S-map f (SX x) = SX (f x)
S-map f Scp    = Scp
S-map f (S⊗ s o) = S⊗ (S-map f s) (S-map f o)
S-map f (Si1 s) = Si1 (S-map f s)
S-map f (Si2 s) = Si2 (S-map f s)

```

Computing a spine is easy, first we check whether or not x and y are equal. If they are, we are done. If not, we inspect the first constructor and traverse it. Then we repeat.

```

mutual
  spine-cp : {ty : U} → [ ty ] A → [ ty ] A → S Δ ty
  spine-cp {ty} x y
    with dec-eq _?_ A_ ty x y
  ...| no _ = spine x y
  ...| yes _ = Scp

  spine : {ty : U} → [ ty ] A → [ ty ] A → S Δ ty
  spine {ty ⊗ tv} (x1 , x2) (y1 , y2)
    = S⊗ (spine-cp x1 y1) (spine-cp x2 y2)
  spine {tv ⊕ tw} (i1 x) (i1 y) = Si1 (spine-cp x y)
  spine {tv ⊕ tw} (i2 x) (i2 y) = Si2 (spine-cp x y)
  spine {ty} x y = SX (delta {ty} {ty} x y)

```

The “application” relations specified by a spine $s = \text{spine-cp } x \ y$, denoted s^b are:

$$\begin{aligned}
\text{Scp}^b &= A \xleftarrow{id} A \\
(\text{S}\otimes s_1 \ s_2)^b &= A \times B \xleftarrow{s_1^b \times s_2^b} A \times B \\
(\text{Si1 } s)^b &= A + B \xleftarrow{i_1 \cdot s^b \cdot i_1^\circ} A + B \\
(\text{Si2 } s)^b &= A + B \xleftarrow{i_2 \cdot s^b \cdot i_2^\circ} A + B \\
(\text{SX } (x, y))^b &= A \xleftarrow{y \cdot x^\circ} A
\end{aligned}$$

Note that, in the $(\text{SX } (x, y))$ case, we already assume there is a Δ there because s was the output of **spine-cp**. In the general case this is slightly different. The **Domain** modules are the ones responsible for handling the “application” relations.

Needs discussion:

Non-canoncity can be a problem: $\text{Scp}^b \equiv (\text{S} \otimes \text{Scp} \text{Scp})^b$ Even though the `spine-cp` function will never find the right-hand above, it feels sub-optimal to allow this.

One possible solution could be to remove `Scp` and handle them through the maybe monad. Instead of $(\text{S } \Delta)$ we would have $(\text{S } (\text{Maybe } \Delta))$, where the `nothings` represent copy. This ensures that we can only copy on the leaves. Branch `explicit-cpy` of the repo has this experiment going. It is easier said than done.

Ignoring the problems and moving forward; note that for any x and y , a spine $s = \text{spine-cp } x \ y$ will NEVER contain a product nor a unit on a leaf (we force going through products and copying units). Hence, whenever we are traversing s and find a `SX`, we know that: (1) the values of the pair are different and (2) we must be at a coproduct, a constant type or a type variable. The constant type or the type variable are out of our control. But we can refine our *description* in case we arrive at a coproduct.

2.3 Coproduct Changes

Let's imagine we are diffing the following values of a type $\mathbb{1} + (\mathbb{N}^2 + \mathbb{N}) \times \mathbb{N}$:

$$\begin{aligned} s &= \text{spine-cp } (\text{i2 } (\text{i1 } (4, 10), 5)) (\text{i2 } (\text{i2 } 10, 5)) \\ &= \text{Si2 } (\text{S} \otimes (\text{SX } (\text{i1 } (4, 10), \text{i2 } 10)) \text{Scp}) \end{aligned}$$

And now, we want to `S-map` over s and refine the Δ inside. We want to have information about which injections we need to pattern match on and which we need to introduce. We begin with a type (that's also a free monad) that encodes the injections we need to insert or pattern-match on:

```
data C (P : UUSet) : U → U → Set where
  CX  : {ty tv : U} → P ty tv → C P ty tv
  Ci1 : {ty tv k : U} → C P ty tv → C P ty (tv ⊕ k)
  Ci2 : {ty tv k : U} → C P ty tv → C P ty (k ⊕ tv)
  Ci1° : {ty tv k : U} → C P ty tv → C P (ty ⊕ k) tv
  Ci2° : {ty tv k : U} → C P ty tv → C P (k ⊕ ty) tv
```

Note that `C` is also a functor on P :

```
C-map : {ty tv : U} {P Q : UUSet}
  → (f : ∀ {k v} → P k v → Q k v)
  → C P ty tv → C Q ty tv
C-map f (CX x) = CX (f x)
C-map f (Ci1 s) = Ci1 (C-map f s)
C-map f (Ci2 s) = Ci2 (C-map f s)
C-map f (Ci1° s) = Ci1° (C-map f s)
C-map f (Ci2° s) = Ci2° (C-map f s)
```

And we make an eager function that will consume ALL coproduct injections from both source and destination:

```
change : {ty tv : U} → [ ty ] A → [ tv ] A → C Δ ty tv
change {ty} {tv ⊕ tw} x (i1 y) = Ci1 (change x y)
change {ty} {tv ⊕ tw} x (i2 y) = Ci2 (change x y)
change {ty ⊕ tw} {tv} (i1 x) y = Ci1° (change x y)
change {ty ⊕ tw} {tv} (i2 x) y = Ci2° (change x y)
change {ty} {tv} x y = CX (delta {ty} {tv} x y)
change-list : {ty tv : U} → [ ty ] A → [ tv ] A → List (C Δ ty tv)
change-list x = return ∘ change x
```

Now, we could `S-map change` over s :

$$\begin{aligned} \text{S-map change } s &= \text{Si2 } (\text{S} \otimes (\text{SX } (\text{Ci2 } (\text{Ci1}^\circ (\text{CX } ((4, 10), 10)))) \text{Scp}) \\ &= s' \end{aligned}$$

The whole thing also becomes of a much more expressive type:

$$s' = \text{S-map change } s : \text{S } (\text{C } \Delta)$$

We can read the type as: a common prefix from both terms followed by injections into the target term and pattern matching on the source term followed by pointwise changes from source to destination.

Note that we can also say, for sure, that we will never have a $\text{SX } (\text{Ci1 } (\text{Ci1}^\circ c))$ at the S-leaves of s' ; this would mean that we need to inject the target with ι_1 and pattern match the source on ι_1 , but this is a common-prefix! Hence it would be recognized by `spine-cp` and instead we would have: $\text{Si1 } (\text{SX } c)$.

Needs discussion:

Even though the algorithm does not produce $\text{SX } (\text{Ci1 } (\text{Ci1}^\circ c))$, as mentioned above, this is still a valid inhabitant of $\text{S } (\text{C } \Delta)$!

Just like the values of S , we can also define the “application” relation induced by the values of C :

$$\begin{aligned} (\text{Ci1 } c)^b &= B + X \xleftarrow{\iota_1 \cdot c^b} A \\ (\text{Ci2 } c)^b &= X + B \xleftarrow{\iota_2 \cdot c^b} A \\ (\text{Ci1}^\circ c)^b &= B \xleftarrow{c^b \cdot \iota_1^\circ} A + X \\ (\text{Ci2}^\circ c)^b &= B \xleftarrow{c^b \cdot \iota_2^\circ} X + A \\ (\text{CX } (x, y))^b &= B \xleftarrow{y \cdot x^\circ} A \end{aligned}$$

Note that up until now, everything was deterministic! This is something we are about to lose.

2.4 Aligning Everything

Following a similar reasoning as from S to C ; the leaves of a C produced through `change` will NEVER contain a coproduct as the topmost type. Hence, we know that they will contain either a product, or a constant type, or a type variable. In the case of a constant type or a type variable, there is not much we can do at the moment, but for a product we can refine this a little bit more before using Δ^1 .

Looking at our running example, we have a leaf $(\text{CX } ((4, 10), 10))$ to take care of. Here the source type is \mathbb{N}^2 and the destination is \mathbb{N} . Note that the \mathbb{N} is treated as a constant type here. As we mentioned above, we have a product on the source, so we could extract some more information before giving up and using Δ !

Here is where our design space starts to be huge. Our definition of alignment is:

¹In fact, splitting the different stages of the algorithm into different types reinforced our intuition that the alignment is the source of difficulties. As we shall see, we now need to introduce non-determinism.

```

data AI (P : UUSet) : U → U → Set where
  AX   : {ty tv : U} → P ty tv → AI P ty tv
  Ap1  : {ty tv tw : U} →  $\llbracket tw \rrbracket A \rightarrow AI P ty tv \rightarrow AI P ty (tv \otimes tw)$ 
  Ap1° : {ty tv tw : U} →  $\llbracket tw \rrbracket A \rightarrow AI P ty tv \rightarrow AI P (ty \otimes tw) tv$ 
  Ap2  : {ty tv tw : U} →  $\llbracket tw \rrbracket A \rightarrow AI P ty tv \rightarrow AI P ty (tw \otimes tv)$ 
  Ap2° : {ty tv tw : U} →  $\llbracket tw \rrbracket A \rightarrow AI P ty tv \rightarrow AI P (tw \otimes ty) tv$ 
  A $\otimes$  : {ty tv tw tz : U}
        → AI P ty tw → AI P tv tz → AI P (ty  $\otimes$  tv) (tw  $\otimes$  tz)

```

Which states that we can force components of the source or destination product to be equal to a given value or we can join two alignments together (This is a big source of inefficiency!).

Computing alignments is very expensive! In the case we actually have products on both the source and the destination we have a lot of options (hence the list monad!):

```

align : {ty tv : U} →  $\llbracket ty \rrbracket A \rightarrow \llbracket tv \rrbracket A \rightarrow \text{List } (AI \Delta ty tv)$ 
align {ty  $\otimes$  ty} {tv  $\otimes$  tv} (x1 , x2) (y1 , y2)
  = A $\otimes$  <$> align x1 y1 <*> align x2 y2
  ++ Ap1 y2 <$> align (x1 , x2) y1
  ++ Ap2 y1 <$> align (x1 , x2) y2
  ++ Ap1° x2 <$> align x1 (y1 , y2)
  ++ Ap2° x1 <$> align x2 (y1 , y2)

```

If we only have products on the left or on the right (or none), we have less options:

```

align {ty  $\otimes$  ty} {tv} (x1 , x2) y
  = Ap1° x2 <$> align x1 y
  ++ Ap2° x1 <$> align x2 y
align {ty} {tv  $\otimes$  tv} x (y1 , y2)
  = Ap1 y2 <$> align x y1
  ++ Ap2 y1 <$> align x y2
align {ty} {tv} x y = return (AX (x , y))

```

Following our previous example, we could **C-mapM** our alignment function (the **C-mapM** is the monadic variant of **C-map**) on s' , in order to find all alignments of (4, 10) and 10. In this case, this is very easy².

```

C-map align s' = [ Si2 (S $\otimes$  (SX (Ci2 (Ci1° (CX (Ap1° 10 (AX (4, 10)))))) Scp)
  , Si2 (S $\otimes$  (SX (Ci2 (Ci1° (CX (Ap2° 4 (AX (10, 10)))))) Scp) ]

```

If we ommit the **SX** and **CX** constructors this becomes slightly more readable:

C-map align s' = [Si2 (S \otimes (Ci2 (Ci1[°] (Ap1[°] 10 (AX (4, 10)))))) S_{cp}) (P1)

, Si2 (S \otimes (Ci2 (Ci1[°] (Ap2[°] 4 (AX (10, 10)))))) S_{cp})] (P2)

But now we end up having to choose between one of those to be *the* patch. This is where we start to need a cost function. Before talking about cost, let's look at the “application” relations of **AI**:

²It might be hard to build intuition for why we need the **A \otimes** constructor. On the **Lab** module of Fixpoint there is an example using 2-3-Trees that motivates the importance of that constructor

$$\begin{aligned}
(\mathbf{A} \otimes a_1 a_2)^b &= B \times D \xleftarrow{a_1^b \times a_1^b} A \times C \\
(\mathbf{Ap1} \ x \ c)^b &= B \times X \xleftarrow{\langle c^b, \underline{x} \rangle} A \\
(\mathbf{Ap2} \ x \ c)^b &= X \times B \xleftarrow{\langle \underline{x}, c^b \rangle} A \\
(\mathbf{Ap1}^\circ \ x \ c)^b &= B \xleftarrow{\pi_1 \cdot (c^b \times \underline{x}^\circ)} A \times X \\
(\mathbf{Ap2}^\circ \ x \ c)^b &= B \xleftarrow{\pi_2 \cdot (\underline{x}^\circ \times c^b)} X \times A \\
(\mathbf{AX} \ (x, y))^b &= B \xleftarrow{\underline{y} \cdot \underline{x}^\circ} A
\end{aligned}$$

3 Patches as Relations

On Section 2.2 we mentioned that the copy constructor was problematic. Another motivation for removing it and handling everything externally is that we would like to say that the patch (P2) above copies the 10, instead of saying that 10 changes into 10. We can postpone this by changing the relation semantics of Δ . We can say that: $(x, x)^b = id$ and $(x, y)^b = \underline{y} \cdot \underline{x}^\circ$.

Nevertheless, if we look at the two patches above as relations, we have:

$$\begin{aligned}
P1^b &= i_2 \cdot (i_2 \cdot \langle \underline{4} \cdot \underline{10}^\circ, \underline{10} \rangle^\circ \cdot i_1^\circ \times id) \cdot i_2^\circ \\
P2^b &= i_2 \cdot (i_2 \cdot \langle \underline{4}, id \rangle^\circ \cdot i_1^\circ \times id) \cdot i_2^\circ
\end{aligned}$$

Writing them in a diagram:

$$\begin{array}{c}
\begin{array}{ccc}
\mathbb{N} & & \\
id \downarrow & \swarrow \pi_1 & \\
\mathbb{N} & & (\mathbb{N}^2 + \mathbb{N}) \times \mathbb{N} \xleftarrow{i_2^\circ} \mathbb{1} + (\mathbb{N}^2 + \mathbb{N}) \times \mathbb{N} \\
& \swarrow \pi_1^\circ & \searrow \pi_2 \\
& & (\mathbb{N}^2 + \mathbb{N}) \times \mathbb{N} \xrightarrow{i_2} \mathbb{1} + (\mathbb{N}^2 + \mathbb{N}) \times \mathbb{N}
\end{array} \\
\begin{array}{ccc}
\mathbb{N}^2 \xleftarrow{i_1^\circ} \mathbb{N}^2 + \mathbb{N} & & \\
\downarrow \langle \underline{4}, id \rangle^\circ & \searrow \pi_2^\circ & \\
\mathbb{N} \xrightarrow{i_2} \mathbb{N}^2 + \mathbb{N} & &
\end{array}
\end{array}$$

$\begin{array}{c} \mathbb{1} \\ \parallel \\ P2^b \\ \parallel \\ P1^b \end{array}$

Here, we have something curious going on... We have that $P1^b \subseteq P2^b$. To see this is not very hard. First, composition and converses are monotonous with respect to \subseteq . We are left to check that:

$$\begin{aligned}
&\langle \underline{4} \cdot \underline{10}^\circ, \underline{10} \rangle \subseteq \langle \underline{4}, id \rangle \\
&\equiv \{ \text{split universsal} \} \\
&\pi_1 \cdot \langle \underline{4} \cdot \underline{10}^\circ, \underline{10} \rangle \subseteq \underline{4} \wedge \pi_2 \cdot \langle \underline{4} \cdot \underline{10}^\circ, \underline{10} \rangle \subseteq id
\end{aligned}$$

The first proof obligation is easy to calculate with:

$$\begin{aligned}
& \pi_1 \cdot < \underline{4} \cdot \underline{10}^\circ, \underline{10} > \subseteq \underline{4} \\
& \Leftarrow \{ \pi_1\text{-cancel} ; \subseteq\text{-trans} \} \\
& \underline{4} \cdot \underline{10}^\circ \subseteq \underline{4} \\
& \Leftarrow \{ \text{Leibniz} \} \\
& \underline{4} \cdot \underline{10}^\circ \cdot \underline{10} \subseteq \underline{4} \cdot \underline{10} \\
& \equiv \{ \underline{a}^\circ \cdot \underline{a} \equiv \top \} \\
& \underline{4} \cdot \top \subseteq \underline{4} \cdot \underline{10} \\
& \equiv \{ \underline{a} \cdot \underline{b} \equiv \underline{a} \} \\
& \underline{4} \cdot \top \subseteq \underline{4} \\
& \equiv \{ \underline{a} \cdot \top \equiv \underline{a} \} \\
& \underline{4} \subseteq \underline{4} \\
& \equiv \{ \subseteq\text{-refl} \} \\
& \text{True}
\end{aligned}$$

The second is easier to prove once we add variables!

$$\begin{aligned}
& \pi_2 \cdot < \underline{4} \cdot \underline{10}^\circ, \underline{10} > \subseteq \text{id} \\
& \equiv \{ \text{Add variables} \} \\
& \forall x, y . x (\pi_2 \cdot < \underline{4} \cdot \underline{10}^\circ, \underline{10} >) y \Rightarrow x = y \\
& \equiv \{ \text{PF expand composition} \} \\
& \forall x, y . \exists z . x (\pi_2) z \wedge z < \underline{4} \cdot \underline{10}^\circ, \underline{10} > y \Rightarrow x = y \\
& \equiv \{ \text{Types force } z = (z_1, z_2) \} \\
& \forall x, y . \exists z_1, z_2 . x (\pi_2) (z_1, z_2) \wedge (z_1, z_2) < \underline{4} \cdot \underline{10}^\circ, \underline{10} > y \Rightarrow x = y \\
& \equiv \{ \pi_2 \text{ def} \} \\
& \forall x, y . \exists z_1, z_2 . x = z_2 \wedge (z_1, z_2) < \underline{4} \cdot \underline{10}^\circ, \underline{10} > y \Rightarrow x = y \\
& \equiv \{ \text{split def} \} \\
& \forall x, y . \exists z_1, z_2 . x = z_2 \wedge z_1 (\underline{4} \cdot \underline{10}^\circ) y \wedge z_2 (\underline{10}) y \Rightarrow x = y \\
& \equiv \{ \text{points def} \} \\
& \forall x, y . \exists z_1, z_2 . x = z_2 \wedge z_1 = 4 \wedge y = 10 \wedge z_2 = 10 \Rightarrow x = y \\
& \equiv \{ \text{substitutions ; weakenings} \} \\
& \forall x, y . \exists z_2 . x = 10 \wedge y = 10 \Rightarrow x = y \\
& \equiv \{ \text{trivial} \} \\
& \text{True}
\end{aligned}$$

Nevertheless, it is clear which patch we should choose! We should always choose the patch that gives rise to the biggest relation, as this is applicable to much more elements.

This suggests an interesting justification for the cost function. For some reason, looks like we won the lottery with our cost functions. We are always choosing the patch that gives rise to the maximal relation. I still don't clearly understand why or how, but it works.

Below are our cost functions:

$$\begin{aligned}
& \text{cost-}\Delta : \{ty \text{ } tv : \mathbb{U}\} \rightarrow \Delta \text{ } ty \text{ } tv \rightarrow \mathbb{N} \\
& \text{cost-}\Delta \{ty\} \{tv\} (x, y) \text{ with U-eq } ty \text{ } tv \\
& \text{cost-}\Delta \{ty\} \{ty\} (x, y) \mid \text{yes refl} \\
& \quad \text{with dec-eq } \underline{\quad} \stackrel{?}{=} \text{-A-} \underline{\quad} ty \text{ } x \text{ } y \\
& \dots \mid \text{yes } \underline{\quad} = 0 \\
& \dots \mid \text{no } \underline{\quad} = \text{size1 sized } ty \text{ } x + \text{size1 sized } ty \text{ } y \\
& \text{cost-}\Delta \{ty\} \{tv\} (x, y) \mid \text{no } \underline{\quad} \\
& \quad = \text{size1 sized } ty \text{ } x + \text{size1 sized } tv \text{ } y
\end{aligned}$$

$\text{S-cost} : \{ty : \mathbf{U}\} \{P : \mathbf{UUSet}\}$
 $\rightarrow (\text{cost} P : \forall \{ty\} \rightarrow P \text{ ty ty} \rightarrow \mathbb{N})$
 $\rightarrow \mathbf{S} P \text{ ty} \rightarrow \mathbb{N}$
 $\text{S-cost } c (\mathbf{SX} \ x) = c \ x$
 $\text{S-cost } c \ \mathbf{Scp} = 0$
 $\text{S-cost } c (\mathbf{S} \otimes s \ o) = \text{S-cost } c \ s + \text{S-cost } c \ o$
 $\text{S-cost } c (\mathbf{Si1} \ s) = \text{S-cost } c \ s$
 $\text{S-cost } c (\mathbf{Si2} \ s) = \text{S-cost } c \ s$

$\text{C-cost} : \{ty \ tv : \mathbf{U}\} \{P : \mathbf{UUSet}\}$
 $\rightarrow (\text{cost} P : \forall \{ty \ tv\} \rightarrow P \text{ ty tv} \rightarrow \mathbb{N})$
 $\rightarrow \mathbf{C} P \text{ ty tv} \rightarrow \mathbb{N}$
 $\text{C-cost } c (\mathbf{CX} \ x) = c \ x$
 $\text{C-cost } c (\mathbf{Ci1} \ s) = 1 + \text{C-cost } c \ s$
 $\text{C-cost } c (\mathbf{Ci2} \ s) = 1 + \text{C-cost } c \ s$
 $\text{C-cost } c (\mathbf{Ci1}^\circ \ s) = 1 + \text{C-cost } c \ s$
 $\text{C-cost } c (\mathbf{Ci2}^\circ \ s) = 1 + \text{C-cost } c \ s$

$\text{Al-cost} : \{ty \ tv : \mathbf{U}\} \{P : \mathbf{UUSet}\}$
 $\rightarrow (\text{cost} P : \forall \{ty \ tv\} \rightarrow P \text{ ty tv} \rightarrow \mathbb{N})$
 $\rightarrow \mathbf{Al} P \text{ ty tv} \rightarrow \mathbb{N}$
 $\text{Al-cost } c (\mathbf{AX} \ xy) = c \ xy$
 $\text{Al-cost } c (\mathbf{A} \otimes s \ o) = \text{Al-cost } c \ s + \text{Al-cost } c \ o$
 $\text{Al-cost } c (\mathbf{Ap1} \ \{tw = k\} \ x \ s) = \text{size1 sized } k \ x + \text{Al-cost } c \ s$
 $\text{Al-cost } c (\mathbf{Ap2} \ \{tw = k\} \ x \ s) = \text{size1 sized } k \ x + \text{Al-cost } c \ s$
 $\text{Al-cost } c (\mathbf{Ap1}^\circ \ \{tw = k\} \ x \ s) = \text{size1 sized } k \ x + \text{Al-cost } c \ s$
 $\text{Al-cost } c (\mathbf{Ap2}^\circ \ \{tw = k\} \ x \ s) = \text{size1 sized } k \ x + \text{Al-cost } c \ s$

3.1 Patches for Regular Types

Now that we have *spines*, *changes* and *alignments* figured out, we can define a patch as:

$\text{Patch} : \mathbf{U} \rightarrow \mathbf{Set}$
 $\text{Patch } ty = \mathbf{S} (\mathbf{C} (\mathbf{Al} \ \Delta)) \ ty$

Computing inhabitants of such type is done with:

$\text{diff1}^* : \{ty : \mathbf{U}\} \rightarrow \llbracket ty \rrbracket A \rightarrow \llbracket ty \rrbracket A \rightarrow \mathbf{List} (\text{Patch } ty)$
 $\text{diff1}^* \ x \ y = \mathbf{S-mapM} (\mathbf{C-mapM} (\text{uncurry align}) \circ \text{uncurry change}) (\text{spine-cp } x \ y)$

4 Mutually Recursive Types

Now that we have a clear picture of regular types, extending this to recursive types is not very difficult.

First, recall that a mutually recursive family is defined as n codes that each reference n type variables:

$\mathbf{Fam} : \mathbb{N} \rightarrow \mathbf{Set}$
 $\mathbf{Fam} \ n = \mathbf{Vec} (\mathbf{U}_n \ n) \ n$
 $\text{data } \mathbf{Fix} \ \{n : \mathbb{N}\} (F : \mathbf{Fam} \ n) : \mathbf{Fin} \ n \rightarrow \mathbf{Set} \text{ where}$
 $\langle _ \rangle : \forall \{k\} \rightarrow \llbracket \text{lookup } k \ F \rrbracket (\mathbf{Fix} \ F) \rightarrow \mathbf{Fix} \ F \ k$

Another auxiliary definition we use here is the indexed coproduct, which let's us *extend* some indexed type.

$_ +_u _ : \{n : \mathbb{N}\} \rightarrow (\mathbf{U}_n \ n \rightarrow \mathbf{U}_n \ n \rightarrow \mathbf{Set}) \rightarrow (\mathbf{U}_n \ n \rightarrow \mathbf{U}_n \ n \rightarrow \mathbf{Set}) \rightarrow (\mathbf{U}_n \ n \rightarrow \mathbf{U}_n \ n \rightarrow \mathbf{Set})$
 $(P +_u Q) \ ty \ tv = (P \ ty \ tv) \uplus (Q \ ty \ tv)$

Now, we already have the ingredients for common prefixes, coproducts and products. We now need to handle type variables. Before we proceed with the nasty definitions, we still need two last synonyms:

```

Fami : Set
Fami = Fin fam#

T : Fami → Un fam#
T k = lookup k fam

```

Here $T\ k$ represents the k -th type of the family, and Fam_i represents the indexes of our family.

Remember that we have **Spines**, to handle the common prefixes; **Changes** to handle different coproduct injections on source and destination and **Alignments** to make sure we exploit products. We kept mentioning that we could not do anything on the leaves that were type variables or constant types. Well, constant types we still can't, and we have to stick to Δ , but now we need to handle the type-variables.

A patch for a fixed point might not follow the precise order of operations (**S**, then **C**, then **Al**) that regular types enjoyed. For instance, imagine we are transforming the following lists:

$$[5, 8, 13, 21] \rightsquigarrow [8, 13, 21]$$

Let lists be seen (as usual) as the initial algebras of $L_A\ X = 1 + A \times X$; then, both lists are inhabitants of μL_N , but, more precisely, the source is an inhabitant of $L_N(L_N(L_N(L_N\ 1)))$ whereas the target is an inhabitant of $L_N(L_N(L_N\ 1))$.

Here, we are already beginning with different types, so a spine (which is homogeneous) might not be the best start! In fact, the best start is to say that the first 5 is deleted, then the spine can kick in and say that everything else is copied!

Here is the definition:

```

data Patchμ : U → U → Set where
  skel : {ty : U} → S Patchμ ty ty
  chng : {ty tv : U} → Cμ (Al Patchμ) ty tv → Patchμ ty tv
  fix : {k k' : Fami} → Patchμ (T k) (T k') → Patchμ (l k) (l k')
  set : {ty tv : U} → Δ ty tv → Patchμ ty tv

```

The **skel** constructor lets a spine kick in, which has **Patchμ** on it's leaves again. The **fix** constructor ties the know between type-variables and their lookup in the family, which is isomorphic. The **set** constructor specifies that we are setting something. The **Cμ** type extends the previous **C** with options for inserting and deleting:

```

data Cμ (P : UUSet) : U → U → Set where
  Cins : {k k' : Fami} → C P (l k) (T k') → Cμ P (T k) (T k')
  Cdel : {k k' : Fami} → C P (T k) (l k') → Cμ P (T k) (T k')
  Cmod : {ty tv : U} → C P ty tv → Cμ P ty tv

```

Note that **fix**, **Cins** and **Cdel** are heterogeneous on the Fam_i indexes! This is very important for mutually recursive families.

Needs discussion:

I have been experimenting with different cost models and slight variations on the definitions of **Patchμ**. I found that requiring **set** to set the *same* type is a great idea! Not only it removes a lot of absurd patches, but it also speeds up the process quite a bit:

```

set : {ty : U} → Δ ty ty → Patchμ ty ty

```

Computing a **Patchμ** is done by piggybacking on the functions for computing **S**, **C** and **Al** separately, then mapping over them with some refinement functions:

```

changeμ : {ty tv : U}
  → [ ty ] (Fix fam) → [ tv ] (Fix fam)
  → List (Cμ (Al Patchμ) ty tv)
changeμ x y = Cmod <$> C-mapM refine-C (change x y)

spineμ : {ty tv : U} → [ ty ] (Fix fam) → [ tv ] (Fix fam) → List (Patchμ ty tv)
spineμ {ty} {tv} x y with U-eq ty tv
...| no _ = chng <$> changeμ x y
spineμ {ty} {ty} x y
  | yes refl = skel <$> S-mapM refine-S (spine-cp x y)

{-# TERMINATING #-}
diffμ* : {k k' : Fami} → Fix fam k → Fix fam k' → List (Patchμ (T k) (T k'))
diffμ* {k} {k'} x y
  = spineμ {T k} {T k'} x y
  ++ ((chng ∘ Cdel {k = k} {k'}) <$> (C-mapM refine-C (change x (y))))
  ++ ((chng ∘ Cins {k = k} {k'}) <$> (C-mapM refine-C (change (x) y)))

```

The refinement functions are given by:

```

mutual
  refine-S : {ty : U} → Δ ty ty → List (Patchμ ty ty)
  refine-S {l k} (x , y) = fix <$> diffμ* x y
  refine-S (x , y) = chng <$> changeμ x y

  refine-C : {k v : U} → Δ k v → List (Al Patchμ k v)
  refine-C {l k} {l k'} (x , y) = (AX ∘ fix) <$> diffμ* x y
  refine-C (x , y) = align x y »= Al-mapM refine-Al

  refine-Al : {k v : U} → Δ k v → List (Patchμ k v)
  refine-Al {l k} {l k'} (x , y) = fix <$> diffμ* x y
  refine-Al (x , y) = return (set (x , y))

```

Obviously, we can also define the “application” relation for fixed points, and that is done in `RegDiff/Diff/Multirec/Domain`. I believe it is more worthwhile to look at some example patches and their “application” relation instead of the general case, though. Let’s begin with the lists we just discussed:

```

s0 : Patchμ LIST-F LIST-F
s0 = diffμ (5 > 8 > 13 > 21 > #) (8 > 13 > 21 > #)

s0-norm : Patchμ LIST-F LIST-F
s0-norm = chng (Cdel (Ci2° (CX (Ap2° 5 (AX (fix (skel Scp)))))))

```

Here, `LIST-F` is defined as `u1⊕N⊗l`, the usual list functor. The “application” relation for the above patch is:

$$\begin{array}{ccc}
 1 + \mathbb{N} \times [\mathbb{N}] & \xrightarrow{\quad s0 \quad} & 1 + \mathbb{N} \times [\mathbb{N}] \\
 \downarrow \iota_2^\circ & & \uparrow \text{in}^\circ \\
 \mathbb{N} \times [\mathbb{N}] & \xrightarrow{\quad \bar{5}^\circ \times id \quad} \mathbb{N} \times [\mathbb{N}] \xrightarrow{\quad \pi_2 \quad} & [\mathbb{N}]
 \end{array}$$

4.1 Examples

Here we add some more examples of patches over fixpoints. These can be seen in the respective `Lab.agda` modules. Here are a few examples of list patches:

```

l0 l1 : list
l0 = (3 > 50 > 4 > #)
l1 = (1 > 50 > 4 > 20 > #)

s1 : Patch $\mu$  LIST-F LIST-F
s1 = diff $\mu$  l0 l1

s1-normalized : Patch $\mu$  LIST-F LIST-F
s1-normalized
= skel
  (Si2
    (S $\otimes$  (SX (chng (Cmod (CX (AX (set (3 , 1)))))))
    (SX
      (fix
        (skel
          (Si2
            (S $\otimes$  Scp
              (SX
                (fix
                  (skel
                    (Si2
                      (S $\otimes$  Scp
                        (SX
                          (fix
                            (chng
                              (Cins (Ci2 (CX (Ap2 20 (AX (fix (skel Scp))))))))))))))))))))))))))

```

Previously we also mentioned that *2-3-Trees* were a good motivation for the $A \otimes$ constructor. Here is the actual example (the full code follows to illustrate how modules are imported and used):

```

module Examples2 where
  open import RegDiff.Generic.Konstants
  open import RegDiff.Generic.Fixpoint konstants keqs public
  open import RegDiff.Generic.Eq konstants keqs public

  2-3-TREE-F :  $U_n 1$ 
  2-3-TREE-F =  $u1 \oplus (K \ k\mathbb{N}) \otimes I \otimes I \oplus (K \ k\mathbb{N}) \otimes I \otimes I \otimes I$ 

  2-3-Tree : Set
  2-3-Tree = Fix 2-3-TREE-F

  Leaf : 2-3-Tree
  Leaf =  $\langle i1 \ unit \rangle$ 

  2-Node :  $\mathbb{N} \rightarrow 2-3-Tree \rightarrow 2-3-Tree \rightarrow 2-3-Tree$ 
  2-Node  $n \ l \ r$  =  $\langle i2 \ (i1 \ (n \ , \ l \ , \ r)) \rangle$ 

  3-Node :  $\mathbb{N} \rightarrow 2-3-Tree \rightarrow 2-3-Tree \rightarrow 2-3-Tree \rightarrow 2-3-Tree$ 
  3-Node  $n \ l \ m \ r$  =  $\langle i2 \ (i2 \ (n \ , \ l \ , \ m \ , \ r)) \rangle$ 

  import RegDiff.Diff.Fixpoint.Base konstants keqs
    as DIFF
  open DIFF.Internal 2-3-TREE-F public

  k0 k1 k2 : 2-3-Tree
  k0 = Leaf
  k1 = 2-Node 1 Leaf Leaf
  k2 = 3-Node 5 Leaf Leaf Leaf
  k3 = 3-Node 3 k1 k2 k2

  t1 t2 : 2-3-Tree
  t1 = 2-Node 4 k1 k2
  t2 = 3-Node 5 k1 Leaf k2

```

The patches we calculate are:

```

r1 r2 : Patch $\mu$  2-3-TREE-F 2-3-TREE-F
r1 = diff $\mu$  t1 t2
r2 = diff $\mu$  k1 k3

```

Which are normalized to the following patches. Note that it is the $A \otimes$ in **r1** that lets us copy **k1** and **k2** from the 2-node to the 3-node.

```

r1-normalized : Patch $\mu$  2-3-TREE-F 2-3-TREE-F
r1-normalized
  = skel
    (Si2
      (SX
        (chgng
          (Cmod
            (Ci2
              (Ci1 $^\circ$ 
                (CX
                  (A $\otimes$  (AX (set (4 , 5)))
                    (A $\otimes$  (AX (fix (skel Scp)))
                      (Ap2  $\langle i1 \ unit \rangle$  (AX (fix (skel Scp))))))))))))))

```

```

r2-normalized : Patch $\mu$  2-3-TREE-F 2-3-TREE-F
r2-normalized
= chng
  (Cins
    (Ci2
      (Ci2
        (CX
          (Ap2 3
            (Ap1
              (⟨ i2 (i2 (5 , ⟨ i1 unit ⟩ , ⟨ i1 unit ⟩ , ⟨ i1 unit ⟩)) ⟩ ,
                ⟨ i2 (i2 (5 , ⟨ i1 unit ⟩ , ⟨ i1 unit ⟩ , ⟨ i1 unit ⟩)) ⟩)
              (AX (fix (skel Scp))))))))))

```